

## LISREL with mean structures: Head Start example

In this example (Magidson (1977), Sörbom (1981)) Sörbom used data on 303 white children from the Head Start summer program, consisting of a Head Start sample ( $N = 148$ ) and a matched Control sample ( $N = 155$ ). The correlations, standard deviations, and means are given in the table below. The children were matched on gender and kindergarten attendance but no attempt has been made to match on social status variables.

		Head Start group					S.D.	Means
Variable	Correlations							
$x_1$	1					1.332	3.520	
$x_2$	.441	1				1.281	3.081	
$x_3$	.220	.203	1			1.075	2.088	
$x_4$	.304	.182	.377	1		2.648	5.358	
$y_1$	.274	.265	.208	.084	1	3.764	19.672	
$y_2$	.270	.122	.251	.198	.664	1	2.677	9.562

		Control group					S.D.	Means
Variable	Correlations							
$x_1$	1					1.360	3.839	
$x_2$	.484	1				1.195	3.290	
$x_3$	.224	.342	1			1.93	2.600	
$x_4$	.268	.215	.387	1		3.239	6.435	
$y_1$	.230	.215	.196	.115	1	3.900	20.415	
$y_2$	.265	.297	.234	.162	.635	1	2.719	10.070

The variables used in Sörbom's reanalysis were:

$x_1$  = Mother's education

$x_2$  = Father's education

$x_3$  = Father's occupation

$x_4$  = Family income

$y_1$  = Score on the Metropolitan Readiness Test

$y_2$  = Score on the Illinois Test of Psycholinguistic Abilities

We want to do the following:

- A** Test whether  $x_1$ ,  $x_2$ ,  $x_3$  and  $x_4$  can be regarded as indicators of a single construct  $\xi$  = “Socio-economic status” for both groups. Is the measurement model the same for both groups? Is there a difference in the mean of  $\xi$  between groups?
- B** Assuming that  $y_1$  and  $y_2$  can be used as indicators of another construct  $\eta$  = “cognitive ability”, test whether the same measurement model applies to both groups. Test the hypothesis of no difference in the mean of  $\eta$  between groups.
- C** Estimate the structural equation

$$\eta = \alpha + \gamma\xi + \zeta.$$

Is  $\gamma$  the same for the two groups? Test the hypothesis  $\alpha = 0$ . Interpret the results.

Consider the measurement model for the four socio-economic status indicators  $x_1$ ,  $x_2$ ,  $x_3$  and  $x_4$ . The model assumes that the  $x$ -variables can be accounted for by a single common factor  $\xi$  :

$$x_i^{(g)} = \tau_i + \lambda_i\xi^{(g)} + \delta_i^{(g)}$$

The superscript,  $g$ , is running over the two groups,  $g = 1$  for Head Start children and  $g = 2$  for control children. There is no superscript for  $\tau$  and  $\lambda$ , since we are using the same observed variables in the two groups. Our main interest is in the mean  $\kappa$  of  $\xi$ . Assuming  $E(x_i^{(g)}) = \tau_i + \lambda_i\kappa$ .

If we add a constant  $c$  to  $\kappa$ , this can be compensated for by subtracting  $\lambda_i c$  from  $\tau_i$ . This means that  $\kappa$  and  $\tau_i$  cannot be identified simultaneously or, phrased in other words, there is no definite origin for the construct  $\xi$ . All we can do is estimate differences among groups, i.e., we can specify the mean of  $\xi$  to be zero in the control group, and then  $\kappa$  is the mean difference in socio-economic status between the experimental group and the control group.

The command file for problem **A** can be patterned after the previous example. For problem **A** we get an overall  $\chi^2$  measure of goodness-of-fit of the model equal to 35.9 with 10 degrees of freedom, indicating that the fit of the model is not very good. An examination of the modification indices reveals that there might be a correlation between the errors  $\delta_1$  and  $\delta_2$ , i.e., when the correlation among the observed variables caused by the construct  $\xi$  has been accounted for, there seems to be a correlation left between  $x_1$  and  $x_2$ . This correlation can be interpreted as an indication that parents' education levels correlate more than can be explained by socio-economic status. By adding the covariance  $\theta_{21}$  we get a model with an acceptable fit:  $\chi^2 = 6.5$  with  $df = 8$ . The difference in degrees of freedom from the previous model is two, since we have added two parameters, namely the covariances  $\theta_{21}^{(1)}$  and  $\theta_{21}^{(2)}$  in the two groups. The estimates and their estimated standard errors are given in the table below. It is seen that the groups differ significantly in socio-economic status, the difference being 0.340 with a standard error equal to 0.096. In order to specify the scale of  $\xi$ ,  $\lambda_{11}$  has been fixed to 1.


Magidson (1977) used two cognitive ability tests as criteria: the Metropolitan Readiness test (MRT) and the Illinois Test of psycholinguistic Abilities (ITPA). Magidson made separate analyses for the two tests, but here we will use the two tests to define the construct cognitive ability. This model is the same as before, except that there are only two  $x$ -variables. As a matter of fact, this model has no degrees of freedom, so one can compute the estimates simply by equating the first and second order moments implied by the model to their observed counterparts. In cognitive ability, the Head Start group is also inferior to the Control group in the sense that the estimated mean difference is negative (-0.743). However, the difference is not significant, having a standard error equal to 0.440.

**Table: Estimates for the measurement model for social status (standard error estimates within parentheses)**

	Head Start group	Control group	Common parameters	
$\theta_{11}^{(\delta)}$	1.494 (0.190)	1.554 (0.200)	$\tau_1^{(x)}$	3.848 (0.092)
$\theta_{22}^{(\delta)}$	1.473 (0.183)	1.154 (0.149)	$\tau_2^{(x)}$	3.326 (0.082)
$\theta_{33}^{(\delta)}$	0.715 (0.137)	0.789 (0.164)	$\tau_3^{(x)}$	2.579 (0.091)
$\theta_{44}^{(\delta)}$	4.305 (0.799)	6.942 (1.127)	$\tau_4^{(x)}$	6.463 (0.236)
$\theta_{21}^{(\delta)}$	0.543 (0.143)	0.500 (0.134)	$\lambda_{21}^{(x)}$	0.864 (0.163)
$\kappa$	-0.340 (0.096)	0.0	$\lambda_{31}^{(x)}$	1.384 (0.290)
$\phi$	0.242 (0.092)	0.329 (0.121)	$\lambda_{41}^{(x)}$	3.305 (0.684)

For problem C we use the combined model shown below, where the main focus is on the structural equation

$$\eta^{(g)} = \alpha^{(g)} + \gamma^{(g)} \xi^{(g)} + \zeta^{(g)}.$$



Here  $\xi$  is “socio-economic status (Status) and  $\eta$  is cognitive ability (Ability).

Previously, we could not find an absolute origin for  $\xi$ , now there is no way to find an absolute origin for  $\eta$ . All we can do is compare groups and look at differences. For example, we could fix  $\alpha$  in the Control group to be zero, and then  $\alpha$  in the Head Start group could be interpreted as the effect of the Head Start program when socio-economic status has been controlled for.

The command file (**EX102C.LIS** in the **LISREL Examples** folder) is

```
HEAD START SUMMER PROGRAM: EXPERIMENTALS
DA NI=6 NOBS=149 NG=2
KM FI=EX102.DAT
SD FI=EX102.DAT
ME FI=EX102.DAT
LA FI=EX102.DAT
SE FI=EX102.DAT
MO NX=4 NK=1 NY=2 NE=1 LX=FR LY=FR TY=FR TX=FR AL=FR KA=FR TD=SY
LE
Ability
LK
Status
FR TD 2 1
FI LX 1 LY 1
VA 1 LX 1 LY 1
OU SE TV MI
HEAD START SUMMER PROGRAM: CONTROLS
DA NOBS=156
KM FI=EX102.DAT
SD FI=EX102.DAT
ME FI=EX102.DAT
LA FI=EX102.DAT
SE FI=EX102.DAT
MO LX=IN TX=IN LY=IN TY=IN GA=IN TD=SP AL=FI KA=FI
LE
Ability
LK
Status
OU
```

The  $\chi^2$  for the combined model equals 27.52 with 22 degrees of freedom, so the fit of the model is acceptable. An examination of the  $\gamma$  parameters in the two groups show that they are probably equal, since  $\gamma^{(1)} = 2.296$  and  $\gamma^{(2)} = 2.027$  with estimated standard errors equal to 0.741 and 0.625, respectively. Thus, the final model is a model with the  $\gamma$ 's constrained to be equal. The  $\chi^2$  for this model is 27.64 with 23 degrees of freedom. The difference in  $\chi^2$  for the last two models can be used as a test of the hypothesis that the  $\gamma$ 's are equal. The  $\chi^2$  with one degree of freedom is 0.12, thus we can treat the  $\gamma$ 's as equal. As the slopes are parallel, it is meaningful to talk about  $\alpha$  as a measure of the effect of Head Start. The estimates of the model are given in the table below. There seems to be no significant effect for the Head Start program when controlling for social status, although the inclusion of social class has changed the negative effect to be positive. The estimate of  $\alpha$  is 0.183 with a standard error equal to 9.377.

**Table: Head Start: estimates for the combined model (standard error estimates within parentheses)**

	Head Start group	Control group	Common parameters	
$\theta_{11}^{(\delta)}$	1.417 (0.186)	1.467 (0.194)	$\tau_1^{(x)}$	3.869 (0.094)
$\theta_{22}^{(\delta)}$	1.438 (0.182)	1.085 (0.144)	$\tau_2^{(x)}$	3.339 (0.083)
$\theta_{33}^{(\delta)}$	0.714 (0.125)	0.839 (0.147)	$\tau_3^{(x)}$	2.573 (0.090)
$\theta_{44}^{(\delta)}$	4.619 (0.736)	7.448 (1.059)	$\tau_4^{(x)}$	6.420 (0.228)
$\theta_{11}^{(\varepsilon)}$	6.316 (1.526)	7.321 (1.593)	$\tau_1^{(y)}$	20.357 (0.286)
$\theta_{22}^{(\varepsilon)}$	1.468 (0.982)	1.645 (0.999)	$\tau_2^{(y)}$	10.085 (0.216)
$\theta_{21}^{(\delta)}$	0.481 (0.140)	0.422 (0.129)	$\lambda_{21}^{(x)}$	0.853 (0.143)
$\alpha$	0.183 (0.377)	0.0	$\lambda_{31}^{(x)}$	1.207 (0.221)
$\phi$	0.315 (0.106)	0.405 (0.131)	$\lambda_{41}^{(x)}$	2.756 (0.515)
$\psi$	6.347 (1.474)	6.185 (1.465)	$\lambda_{21}^{(y)}$	0.850 (0.141)
$\kappa$	-0.382 (0.103)	0.0	$\gamma$	2.136 (0.549)

In the more general case, when there are more than two groups, or more than one dependent variable, or both, one can test the hypothesis of no effect by re-estimation of the model with the restriction  $\alpha^{(g)} = 0$  added, and then compare the  $\chi^2$ 's. In the above case we obtain  $\chi^2$  equal to 27.87 with 24 degrees of freedom. Thus the test of no effect results in a  $\chi^2$  with one degree of freedom equal to 0.23, which in this case is the same as the estimate of  $\alpha$  divided by its standard error, squared.